


# Superficies

## Representación de distintos tipos de superficies

### ■ Grafos

```
In[1]:= f[x_, z_] := Sin[x] * z^2 / 7
X = {u, f[u, v], v};
ParametricPlot3D[Evaluate[X], {u, -5, 5}, {v, -5, 5}, PlotPoints -> 30]
```



### ■ Superficies Tubulares

Recordamos que, dada  $\alpha(t)$  curva regular y  $r < \min(R(t))$ , se puede definir la superficie que envuelve de forma tubular a  $\alpha$  como aquella que tiene por parametrización:


$$X(u, v) = \alpha(u) + r \cos(v) N(u) + r \sin(v) T(u).$$

Primero creamos los elementos que necesitamos de la curva  $\alpha$


```
In[4]:= Clear["Global`*"]
norma[z_] := Sqrt[z.z];
alpha = {3 * Cos[u], 3 * Sin[u], u};
T1 = D[alpha, u] / norma[D[alpha, u]] // Simplify;
B1 = Cross[D[alpha, u], D[alpha, {u, 2}]] /
     norma[Cross[D[alpha, u], D[alpha, {u, 2}]]] // Simplify;
N1 =
  Cross[
    B1,
    T1];
```

Veamos (dibujándolo) la evolución del radio de curvatura.

```
In[10]:= curvat = norma[Cross[D[alpha, u], D[alpha, {u, 2}]]] / norma[D[alpha, u]^3] // Simplify;
Plot[Evaluate[1 / curvat], {u, 0, 2 Pi}]
```



```
In[12]:= r = .5;
X = alpha + r * Cos[v] * N1 + r * Sin[v] * B1;
ParametricPlot3D[Evaluate[X], {u, 0, 3 Pi},
{v, 0, 2 Pi}, ViewPoint -> {4.5, -7.5, 1.5}, PlotPoints -> 30]
```



## ■ Superficies de traslación

Partimos de dos curvas,  $\alpha$  y  $\beta$  y buscamos sus puntos de corte.

```
In[15]:= Clear["Global`*"]
a = 2; b = 3;
alpha[u_] := {u, 0, u^2 / a};
beta[v_] := {0, v - 1, -(v - 1)^2 / b}
ss = Solve[alpha[u] == beta[v], {u, v}]
```

Out[19]= {{u -> 0, v -> 1}}



El punto de corte es  $P=\alpha(0)=\beta(1)=(0,0,0)$ .

La parametrización es:

$$X(u,v) = \alpha(u)+\beta(v)-P$$

```
In[20]:= P = alpha[u] /. ss[[1]];
X = alpha[u] + beta[v] - P
ParametricPlot3D[{alpha[u], beta[v]}, {u, -2, 2}, {v, -1, 3}]
ParametricPlot3D[Evaluate[X], {u, -2, 2}, {v, -1, 3}]
```


```
Out[21]= {u, -1 + v,  $\frac{u^2}{2} - \frac{1}{3}(-1 + v)^2$ }
```



## ■ Superficies de revolución

Partimos de una función  $y = f(z) > 0$  y la giramos en torno al eje OZ

```
In[24]:= Clear["Global`*"]
f[z_] := 2 + Cos[z] * Sin[2 z];
X = {Cos[u] * f[v], Sin[u] * f[v], v};
ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, 0, 8}]
```



### ■ Superficies regladas


A partir de curva  $\alpha(u)$ , y dirección generatriz:  $w(u)$  ya tenemos nuestra parametrización:

$$X(u,v) = \alpha(u) + v w(u)$$

Vamos a hacer la del paraboloides hiperbólico (silla de montar)  $z=kxy$ :

```
In[28]:= k = 1 / 4; alpha = {u, 0, 0}; w = {0, 1 / k, u};
X = alpha + v w;
Print["Ecuación de la superficie regladas x(u,v) = ", X]
ParametricPlot3D[Evaluate[X], {u, -5, 5}, {v, -5, 5}, AxesLabel -> {x, y, z}]
```

Ecuación de la superficie regladas  $x(u,v) = \{u, 4 v, u v\}$




### ■ Superficies regladas (conos)

A partir de un vértice  $P$  y una curva  $\alpha(u)$ , tomamos como dirección generatriz:  $w(u) = \alpha(u) - P$  y ya tenemos nuestra parametrización:

$$X(u,v) = \alpha(u) + v w(u) \quad (\text{ó también } X_2(u,v) = p + v w(u)) \quad (\text{véase}).$$

```
In[32]:= Clear["Global`*"]
P = {1, 2, 3};
alpha = {Cos[u], Sin[2 u] Cos[u], 1 - Cos[u] - Sin[u]};
w = alpha - P;
X = P + v w;
Print["Ecuación de la superficie cónica x(u,v) = ", X]
ParametricPlot3D[Evaluate[X], {u, 0, 2 Pi}, {v, -2, 4}, AxesLabel -> {x, y, z}]
```

Ecuación de la superficie cónica  $x(u,v) =$   
 $\{1 + v(-1 + \cos[u]), 2 + v(-2 + \cos[u] \sin[2u]), 3 + v(-2 - \cos[u] - \sin[u])\}$




---

## Geometría de superficies (primera y segunda forma fundamental)

### ■ Primera forma fundamental a partir de una parametrización

Dada  $S = \{X(u,v), (u,v) \in U \subset \mathbb{R}^2\}$  la primera forma fundamental (sus componentes respecto de  $X_u$  y  $X_v$ ) son

$$E = |X_u|^2$$

$$F = \langle X_u, X_v \rangle$$

$$G = |X_v|^2$$

Lo escribimos sobre un ejemplo, el toro (hecho en clase):

```
In[39]:= Clear["Global`*"]
X = {(a + r * Cos[u]) * Cos[v], (a + r * Cos[u]) * Sin[v], r * Sin[u]};
EE = D[X, u].D[X, u] // Simplify;
FF = D[X, u].D[X, v];
GG = D[X, v].D[X, v] // Simplify;
Print[
  "La primera forma fundamental del toro en la parametrización X(u,v) es: \n I(u,v) = ",
  {{EE, FF}, {FF, GG}} // MatrixForm]
```

La primera forma fundamental del toro en la parametrización  $X(u,v)$  es:


$$I(u,v) = \begin{pmatrix} r^2 & 0 \\ 0 & (a + r \cos[u])^2 \end{pmatrix}$$

### ■ Área de una superficie en paramétricas

Partimos de una superficie que se construye mediante combinaciones convexas del segmento  $\overline{ab}$  y el  $\overline{cd}$

```
In[45]:= Clear["Global`*"]
a = {2, 0, 4};
b = {2, 4, 3};
c = {0, 4, 3};
d = {0, 0, 2};
segmento1 = a (1 - u) + b u;
segmento2 = c u + (1 - u) d;
X = Simplify[(1 - v) * segmento1 + v * segmento2];
Print["Parametrización del la superficie x(u,v)=", X]
ParametricPlot3D[Evaluate[X], {u, 0, 1}, {v, 0, 1}]

Parametrización del la superficie x(u,v)={2 - 2 v, 4 u, 4 - 2 v + u (-1 + 2 v)}
```



Calculamos el área desde la definición:  $A = \int |X_u \times X_v| d(uv)$

```
In[55]:= n = Cross[D[X, u], D[X, v]];
area = NIntegrate[Evaluate[Sqrt[n.n]], {u, 0, 1}, {v, 0, 1}]

Out[56]= 9.25525
```

### ■ Área de una superficie pasando por la primera forma fundamental E, F y G

Usamos el mismo ejemplo anterior, determinamos la primera forma fundamental y calculamos el Área aplicando la fórmula

de clase a partir de la parametrización:  $A = \int \sqrt{EG - F^2} d(uv)$

```
In[57]:= EE = D[X, u].D[X, u];
FF = D[X, u].D[X, v];
GG = D[X, v].D[X, v];
area = NIntegrate[Evaluate[Sqrt[EE * GG - FF^2]], {u, 0, 1}, {v, 0, 1}]

Out[60]= 9.25525
```

### ■ Área de una superficie grafo

Dada la superficie (Bóveda de Viviani) cuya ecuación implícita es la de una esfera de radio  $2r$

$$x^2 + y^2 + z^2 = 4r^2$$


pero restringida a  $\{z > 0\}$  y conjunto de valores en el disco:  $U = \{(x, y) \text{ tal que } (x-r)^2 + y^2 < r^2\}$ .

Lo nuevo aquí es parametrizar  $U$ : ya que

$$x \in (1-r, 1+r) \quad y \in \left( -\sqrt{r^2 - (x-r)^2}, \sqrt{r^2 - (x-r)^2} \right),$$

Vamos a representarla para  $r = 1$  como grafo (despejando la  $z$ ) y a calcular su área.

```
In[61]:= Clear["Global`*"]
f[x_, y_] := Sqrt[4 r^2 - x^2 - y^2];
X = {x, y, f[x, y]};
r = 1;
ParametricPlot3D[Evaluate[X], {x, 1 - r, 1 + r},
  {y, -Sqrt[r^2 - (x - r)^2], Sqrt[r^2 - (x - r)^2]}]
```



Y calculamos el Área aplicando la fórmula de clase a partir de la parametrización:  $A = \int \sqrt{1 + |\nabla f|^2} d(xy)$

```
In[66]:= r = 1;
integrando = Sqrt[1 + D[f[x, y], x]^2 + D[f[x, y], y]^2] // Simplify;
area = NIntegrate[Evaluate[integrando],
  {x, 1 - r, 1 + r}, {y, -Sqrt[r^2 - (x - r)^2], Sqrt[r^2 - (x - r)^2]}]
```

Out[68]= 4.56637

### ■ Aplicación de Gauss y segunda forma fundamental a partir de una parametrización

Dada  $S = \{X(u,v), (u,v) \in U \subset \mathbb{R}^2\}$  recordamos que la aplicación de Gauss es la que asigna el vector normal:

$$N = \frac{X_u \times X_v}{|X_u \times X_v|}$$

y las componentes de  $\Pi$  son ( respecto de  $X_u$  y  $X_v$ )

$$e = \langle N, X_{uu} \rangle$$

$$f = \langle N, X_{uv} \rangle$$

$$g = \langle N, X_{vv} \rangle$$

mientras que la matriz de  $-D_p N$  respecto de la base  $B = \{X_u, X_v\}$  es

$$- \begin{pmatrix} E & F \\ F & G \end{pmatrix}^{-1} \begin{pmatrix} e & f \\ f & g \end{pmatrix}$$

Lo aplicamos sobre un paraboloides hiperbólico

```
In[69]:= Clear["Global`*"]
X = {u, v, u^2 - v^2};
n = Cross[D[X, u], D[X, v]];
NN = n / Sqrt[n.n];
Print["Función de Gauss N(u,v) = ", NN]
ee = NN.D[X, {u, 2}] // Simplify;
ff = NN.D[X, u, v];
gg = NN.D[X, {v, 2}] // Simplify;
Print["Matriz de II(u,v) = ", {{ee, ff}, {ff, gg}} // MatrixForm]
```

$$\text{Función de Gauss } N(u,v) = \left\{ -\frac{2u}{\sqrt{1+4u^2+4v^2}}, \frac{2v}{\sqrt{1+4u^2+4v^2}}, \frac{1}{\sqrt{1+4u^2+4v^2}} \right\}$$

$$\text{Matriz de II}(u,v) = \begin{pmatrix} \frac{2}{\sqrt{1+4u^2+4v^2}} & 0 \\ 0 & -\frac{2}{\sqrt{1+4u^2+4v^2}} \end{pmatrix}$$

Las matrices correspondientes a I e II son

```
In[78]:= EE = D[X, u].D[X, u] // Simplify;
FF = D[X, u].D[X, v];
GG = D[X, v].D[X, v] // Simplify;
matriz1 = {{EE, FF}, {FF, GG}};
auxiliar = {{ee, ff}, {ff, gg}};
matriz2 = Inverse[-matriz1].auxiliar;
Print["Matriz de I(u,v) = ", matriz1 // MatrixForm]
Print["Matriz de -DN(u,v) = ", matriz2 // MatrixForm]
```

$$\text{Matriz de I}(u,v) = \begin{pmatrix} 1+4u^2 & -4uv \\ -4uv & 1+4v^2 \end{pmatrix}$$

$$\text{Matriz de -DN}(u,v) = \begin{pmatrix} \frac{2(-1-4v^2)}{(1+4u^2+4v^2)^{3/2}} & \frac{8uv}{(1+4u^2+4v^2)^{3/2}} \\ -\frac{8uv}{(1+4u^2+4v^2)^{3/2}} & \frac{2(-1-4u^2)}{(1+4u^2+4v^2)^{3/2}} \end{pmatrix}$$

## ■ Curvatura de Gauss, curvatura media

Recordamos las definiciones

$-k_1$  y  $k_2$  son, cambiados de signo, los autovalores de la matriz de -DN

$-K$  = curvatura de Gauss =  $\det(\text{matriz -DN}) = k_1 k_2$ ,

$-H$  = curvatura media =  $-\frac{1}{2}\text{traza}(\text{matriz -DN}) = \frac{k_1+k_2}{2}$ .

```
In[86]:= KK = Det[matriz2] // Simplify
HH = -(matriz2[[1, 1]] + matriz2[[2, 2]]) / 2 // Simplify
```

$$\text{Out[86]} = -\frac{4}{(1+4u^2+4v^2)^2}$$

$$\text{Out[87]} = -\frac{4(u^2 - v^2)}{(1+4u^2+4v^2)^{3/2}}$$



```
In[88]:= {k1, k2} = -Eigenvalues[matriz2] // Simplify;
k1
k2
```

$$\text{Out[89]} = \frac{1}{(1 + 4u^2 + 4v^2)^{7/2}} 2 \left( -2u^2 - 16u^4 - 32u^6 + 2v^2 - 32u^4v^2 + 16v^4 + 32u^2v^4 + 32v^6 + \sqrt{(1 + 4u^2 + 4v^2)^4 (4u^4 + u^2(4 - 8v^2) + (1 + 2v^2)^2)} \right)$$

$$\text{Out[90]} = -\frac{1}{(1 + 4u^2 + 4v^2)^{7/2}} 2 \left( 2u^2 + 16u^4 + 32u^6 - 2v^2 + 32u^4v^2 - 16v^4 - 32u^2v^4 - 32v^6 + \sqrt{(1 + 4u^2 + 4v^2)^4 (4u^4 + u^2(4 - 8v^2) + (1 + 2v^2)^2)} \right)$$

## ■ Clasificación de puntos

Recordamos que  $K$  caracteriza los puntos

- $K < 0$  punto hiperbólico (dos direcciones perpendiculares con curvaturas opuestas, tipo silla de montar)
- $K > 0$  punto elíptico (en todas las direcciones, las curvas están “hacia el mismo lado”, tipo elipsoide)
- $K = 0$  punto parabólico (tipo cilindro) que puede ser plano si  $H = 0$  también.

Se dice umbilical cuando  $k_1 = k_2$  (elíptico pero con igual curvatura en todas direcciones, tipo esfera)

```
In[91]:= P = {u -> 1, v -> 1};
Print["Punto P = ", X /. P]
Print["Curvatura de Gauss K en P = ", KK /. P]
Print["Curvatura media H en P = ", HH /. P]
Print["curvatura principal k1 en P = ", k1 /. P]
Print["curvatura principal k2 en P = ", k2 /. P]
```

Punto  $P = \{1, 1, 0\}$

$$\text{Curvatura de Gauss } K \text{ en } P = -\frac{4}{81}$$

Curvatura media  $H$  en  $P = 0$

$$\text{curvatura principal } k_1 \text{ en } P = \frac{2}{9}$$

$$\text{curvatura principal } k_2 \text{ en } P = -\frac{2}{9}$$

En este caso,  $K < 0$ , tenemos un punto no umbilical (hiperbólico), por lo que podemos calcular sus direcciones principales:

## ■ Direcciones principales, indicatriz de Dupin, dirección asintótica

Las direcciones principales son:

```
In[97]:= {v1, v2} = -Eigenvectors[matriz2 /. P];
Print["Dirección principal v1 en P respecto de la base {Xu, Xv}= ", v1]
Print["Dirección principal v1 en P = ", (v1[[1]] * D[X, u] /. P) + (v1[[2]] * D[X, v] /. P)]
Print["Dirección principal v2 en P respecto de la base {Xu, Xv}= ", v2]
Print["Dirección principal v1 en P = ", (v2[[1]] * D[X, u] /. P) + (v2[[2]] * D[X, v] /. P)]
```

Dirección principal  $v_1$  en  $P$  respecto de la base  $\{X_u, X_v\} = \{-2, -1\}$

Dirección principal  $v_1$  en  $P = \{-2, -1, -2\}$

Dirección principal  $v_2$  en  $P$  respecto de la base  $\{X_u, X_v\} = \left\{-\frac{1}{2}, -1\right\}$

Dirección principal  $v_1$  en  $P = \left\{-\frac{1}{2}, -1, 1\right\}$

La indicatriz de Dupin son las cónicas en  $T_p S$  de coordenadas  $(a,b)$  cumpliendo:

$$k_1 a^2 + k_2 b^2 = 1, k_1 a^2 + k_2 b^2 = -1$$

Dejamos como ejercicio el representarla.

```
In[102]:= a^2 (k1 /. P) + b^2 (k2 /. P) == -1
          a^2 (k1 /. P) + b^2 (k2 /. P) == 1
```

```
Out[102]=  $\frac{2 a^2}{9} - \frac{2 b^2}{9} == -1$ 
```

```
Out[103]=  $\frac{2 a^2}{9} - \frac{2 b^2}{9} == 1$ 
```

En este caso,  $K < 0$ , y las direcciones asintóticas (vectores con curvatura normal cero) tienen sentido precisamente en puntos hiperbólicos (o parabólicos).

En términos de la segunda forma es  $\Pi_p(w) = 0$  (en coordenadas respecto de  $\{X_u, X_v\}$  será:

$$\begin{pmatrix} a & b \end{pmatrix} \begin{pmatrix} e & f \\ f & g \end{pmatrix} \begin{pmatrix} a \\ b \end{pmatrix} = 0.$$

```
In[104]:= matrizII = {{ee, ff}, {ff, gg}} /. P;
          asin = Solve[{a, b}.matrizII.{a, b} == 0, {a, b}];
          {a, b} /. asin[[1]]
          {a, b} /. asin[[2]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. >>

```
Out[106]= {-b, b}
```

```
Out[107]= {b, b}
```

En este caso, las direcciones asintóticas son dos (las asíntotas de la indicatriz de Dupin), en las direcciones  $w_1 = (1, -1)$  y  $w_2 = (1, 1)$ . En coordenadas cartesianas, serían (dejamos libre el valor de  $a$ )

```
In[108]:= Print["Dirección asintótica w1 en P = ", a * (D[X, u] /. P) - (a * D[X, v] /. P)]
          Print["Dirección asintótica w2 en P = ", a * (D[X, u] /. P) + (a * D[X, v] /. P)]
```

Dirección asintótica  $w_1$  en  $P = \{a, -a, 4a\}$

Dirección asintótica  $w_2$  en  $P = \{a, a, 0\}$

## ■ Ejercicio

a) Determina una parametrización para la superficie que resulta de rotar la curva

$$\alpha(t) = (2 + \cos(t), \sin(t), t)$$


en torno al eje OZ.

b) Dibújala

c) Verifica que una de sus curvaturas principales no cambia de signo, y la otra sí.

d) Determina un punto elíptico, uno parabólico, uno hiperbólico y uno umbilical.

**Posible solución :**



Punto elíptico  $P = \left\{ \sqrt{5}, 0, \frac{\pi}{2} \right\}$  con curvaturas principales  $-\frac{1}{3}$  y  $-\frac{4}{27}$

Punto hiperbólico  $P = \{1, 0, \pi\}$  con curvaturas principales  $-1$  y  $2$

Punto parabólico  $P = \left\{ \sqrt{3}, 0, -\frac{2\pi}{3} \right\}$  con curvaturas principales  $-\frac{1}{\sqrt{6}}$  y  $0$

Punto umbilical  $P = \left\{ \sqrt{7}, 0, -\frac{\pi}{3} \right\}$  con curvaturas principales  $-\frac{1}{\sqrt{10}}$  y  $-\frac{1}{\sqrt{10}}$

■ **Solución :**